

Online Supplements for:
**Using Bifactor Exploratory Structural Equation Modeling to Test for a Continuum
Structure of Motivation**
(doi:10.1177/0149206316645653)

1. ICM-CFA	Page 2
2. Bifactor CFA	Page 3
3. ESEM	Page 4
4. Bifactor ESEM	Page 5
5. Bifactor ESEM with covariates	Page 7
5. Bifactor ESEM with ESEM-within-CFA	Page 8

Title: 6 Factor CFA;

! In all input files, statements preceded by ! are annotations.

! Use the following statement to identify the data set. Here, the data set is labelled BESEM.dat.

! If the data set is not in the same folder as the input file, include the complete path to the data set.

Data:

File is BESEM.dat;

! The variables names function identifies all variables in the data set, in order of appearance,

! whereas the usevar command identifies the variables used in the analysis.

Variable:

Names are

Am1 Am5 Am6
Esap2 Esap4 Esav1
Emap1 Emap4 Emav4
Inap1 Inap2 Inav1 Inav2
Ident1 Ident3 Ident4
Intrin2 Intrin4 Intrin6;

Usevariable are Am1 Am5 Am6

Esap2 Esap4 Esav1
Emap1 Emap4 Emav4
Inap1 Inap2 Inav1 Inav2
Ident2 Ident3 Ident4
Intrin2 Intrin4 Intrin6;

! The next section defines the analysis. Here the Maximum Likelihood Robust (MLR) estimator is used.

Analysis:

ESTIMATOR = MLR;

! The next statement defines the model. Here, a simple CFA model with no cross loading is specified

! with 6 factors (amotiv - intrin) defined respectively with items from the usevariable list.

! The name of the factors is selected by the user and comes before the "By" command.

! The "By" command indicates with items serve to define which factor.

Model:

Amotiv by Am1 Am5 Am6;
ExtMat by Emap1 Emap4 Emav4;
ExtSoc by Esap2 Esap4 Esav1;
Introj by Inap1 Inap2 Inav1 Inav2;
Ident by Ident2 Ident3 Ident4;
Intrin by Intrin2 Intrin4 Intrin6;

! Specific sections of output are requested

Output: sampstat stdyx mod res svalues;

Title: Bifactor CFA;

! Common sections of inputs are skipped to focus only on changes in the MODEL section
! The next statement defines the model. Here, a bifactor CFA model is
! specified with 6 specific factors (amotiv - intrin) defined as in the CFA model.
! All items are also used to define global factor called G.

Model:

Amotiv by Am1 Am5 Am6;
ExtMat by Emap1 Emap4 Emav4;
ExtSoc by Esap2 Esap4 Esav1;
Introj by Inap1 Inap2 Inav1 Inav2;
Ident by Ident2 Ident3 Ident4;
Intrin by Intrin2 Intrin4 Intrin6;

G by

Am1 Am5 Am6
Esap2 Esap4 Esav1
Emap1 Emap4 Emav4
Inap1 Inap2 Inav1 Inav2
Ident2 Ident3 Ident4
Intrin2 Intrin4 Intrin6;

! All factors are specified as orthogonal, with their correlations (WITH) constrained to be 0 (@0).

Extsoc with amotiv@0;
ExtSoc with Extmat@0;
ExtSoc with Introj@0;
ExtSoc with Ident@0;
ExtSoc with Intrin@0;
Extmat with amotiv@0;
ExtMat with Introj@0;
ExtMat with Ident@0;
ExtMat with Intrin@0;
Introj with amotiv@0;
Introj with Ident@0;
Introj with Intrin@0;
ident with amotiv@0;
Ident with Intrin@0;
Intrin with amotiv@0;
g with amotiv@0;
g with ExtSoc@0;
g with ExtMat@0;
g with Introj@0;
g with Ident@0;
g with Intrin@0;

Output: sampstat stdyx mod res svalues;

TITLE: 6 factor ESEM

! Common sections of inputs are skipped to focus only on the ANALYSIS and MODEL sections.

! The Maximum Likelihood Robust (MLR) estimator is used together with the oblique target rotation.

ANALYSIS:

ESTIMATOR = MLR; ROTATION = target;

! The next statement defines the model. Here, an ESEM model is specified with target rotation.

! The 6 factors (amotiv - intrin) are defined respectively with main loadings from their respective items.

! In addition to these main loadings, all other cross-loadings are estimated but targeted

! to be as close to 0 as possible (~0). Factors forming a single set of ESEM factors (with cross-

*! loadings between factors, are indicated by using the same label in parenthesis after * (*1).*

MODEL:

Amotiv BY Am1 Am5 Am6

Emap1~0 Emap4~0 Emav4~0
Esap2~0 Esap4~0 Esav1~0
Inap1~0 Inap2~0 Inav1~0 Inav2~0
Ident2~0 Ident3~0 Ident4~0
Intrin2~0 Intrin4~0 Intrin6~0 (*1);

Ext_mat by Emap1 Emap4 Emav4

Am1~0 Am5~0 Am6~0
Esap2~0 Esap4~0 Esav1~0
Inap1~0 Inap2~0 Inav1~0 Inav2~0
Ident2~0 Ident3~0 Ident4~0
Intrin2~0 Intrin4~0 Intrin6~0 (*1);

Ext_soc by Esap2 Esap4 Esav1

Am1~0 Am5~0 Am6~0
Emap1~0 Emap4~0 Emav4~0
Inap1~0 Inap2~0 Inav1~0 Inav2~0
Ident2~0 Ident3~0 Ident4~0
Intrin2~0 Intrin4~0 Intrin6~0 (*1);

Introj by Inap1 Inap2 Inav1 Inav2

Am1~0 Am5~0 Am6~0
Emap1~0 Emap4~0 Emav4~0
Esap2~0 Esap4~0 Esav1~0
Ident2~0 Ident3~0 Ident4~0
Intrin2~0 Intrin4~0 Intrin6~0 (*1);

Ident by Ident2 Ident3 Ident4

Am1~0 Am5~0 Am6~0
Emap1~0 Emap4~0 Emav4~0
Esap2~0 Esap4~0 Esav1~0
Inap1~0 Inap2~0 Inav1~0 Inav2~0
Intrin2~0 Intrin4~0 Intrin6~0 (*1);

Intrin by Intrin2 Intrin4 Intrin6

Am1~0 Am5~0 Am6~0
Emap1~0 Emap4~0 Emav4~0
Esap2~0 Esap4~0 Esav1~0

```

    Inap1~0 Inap2~0 Inav1~0 Inav2~0
    Ident2~0 Ident3~0 Ident4~0 (*1);
Output:  sampstat stdyx mod res svalues;
TITLE:  Bifactor-ESEM (measurement model)
! Common sections of inputs are skipped to focus only on the ANALYSIS and MODEL
sections.
! The next section defines the analysis. Here the Maximum Likelihood Robust (MLR)
estimator is used
! together with orthogonal bifactor target rotation (making all factors orthogonal as in
Bifactor-CFA).
ANALYSIS:
ESTIMATOR = MLR; ROTATION = target (orthogonal);
! The next statement defines the model. Here, an ESEM model is specified with target
rotation.
! The 6 factors (amotiv - intrin) are defined respectively with main loadings from their
respective items.
! In addition to these main loadings, all other cross-loadings are estimated but targeted
! to be as close to 0 as possible (~0). All items are also used to define one global factor
(called G).
! Factors forming a single set of ESEM factors (with cross- loadings between factors,
! are indicated by using the same label in parenthesis after * (*1).
MODEL:
G by  Am1 Am5 Am6
     Esap2 Esap4 Esav1
     Emap1 Emap4 Emav4
     Inap1 Inap2 Inav1 Inav2
     Ident2 Ident3 Ident4
     Intrin2 Intrin4 Intrin6 (*1);
Amotiv BY Am1 Am5 Am6
      Emap1~0 Emap4~0 Emav4~0
      Esap2~0 Esap4~0 Esav1~0
      Inap1~0 Inap2~0 Inav1~0 Inav2~0
      Ident2~0 Ident3~0 Ident4~0
      Intrin2~0 Intrin4~0 Intrin6~0 (*1);
Ext_mat by Emap1 Emap4 Emav4
      Am1~0 Am5~0 Am6~0
      Esap2~0 Esap4~0 Esav1~0
      Inap1~0 Inap2~0 Inav1~0 Inav2~0
      Ident2~0 Ident3~0 Ident4~0
      Intrin2~0 Intrin4~0 Intrin6~0 (*1);
Ext_soc by Esap2 Esap4 Esav1
      Am1~0 Am5~0 Am6~0
      Emap1~0 Emap4~0 Emav4~0
      Inap1~0 Inap2~0 Inav1~0 Inav2~0
      Ident2~0 Ident3~0 Ident4~0
      Intrin2~0 Intrin4~0 Intrin6~0 (*1);
Introj by Inap1 Inap2 Inav1 Inav2
      Am1~0 Am5~0 Am6~0
      Emap1~0 Emap4~0 Emav4~0
      Esap2~0 Esap4~0 Esav1~0

```

```
Ident2~0 Ident3~0 Ident4~0
Intrin2~0 Intrin4~0 Intrin6~0 (*1);
Ident by Ident2 Ident3 Ident4
Am1~0 Am5~0 Am6~0
Emap1~0 Emap4~0 Emav4~0
Esap2~0 Esap4~0 Esav1~0
Inap1~0 Inap2~0 Inav1~0 Inav2~0
Intrin2~0 Intrin4~0 Intrin6~0 (*1);
Intrin by Intrin2 Intrin4 Intrin6
Am1~0 Am5~0 Am6~0
Emap1~0 Emap4~0 Emav4~0
Esap2~0 Esap4~0 Esav1~0
Inap1~0 Inap2~0 Inav1~0 Inav2~0
Ident2~0 Ident3~0 Ident4~0 (*1);
Output: sampstat stdyx mod res svalues;
```

TITLE: Bifactor ESEM with covariates relations with all factors freely estimated
! Common sections of inputs are skipped to focus only on the USEVARIABLE, ANALYSIS and MODEL sections. Variables autonomy, competence, relatedness, affective commitment (AC), and continuance commitment (CC) have been added to this analysis in the use variable section.
Usevariable are
Am1 Am5 Am6 Emap1 Emap4 Emav4
Esap2 Esap4 Esav1 Inap1 Inap2 Inav1 Inav2
Ident2 Ident3 Ident4 Intrin2 Intrin4 Intrin6
Autonomy Competence Relatedness AC CC ;
! The next section defines the analysis. Here the Maximum Likelihood Robust (MLR) estimator is used together with the orthogonal bifactor target rotation.
Analysis:
ESTIMATOR = MLR; ROTATION = target (orthogonal);
Model:
![...] The first section of the model is identical to the previous example.
! The following ON statement specifies estimation of regressions between S-factors (amotiv – to intrin) and G-factor and covariates (Autonomy, Competence, Relatedness, AC, and CC).
AC CC Autonomy Competence Relatedness ON Amotiv ExtMat ExtSoc Introj Ident Intrin
G ;

TITLE: Bifactor ESEM with ESEM-within-CFA with covariates relations with the S-factors constrained to be zero

! ANALYSIS and MODEL sections only

Analysis: estimator is MLR;

! Start values in the following section are obtained through estimation of the bifactor-ESEM measurement

! model requesting the "svalues" section of the output.

! Using these values is required to relax some constraints of ESEM and Bifactor-ESEM. For instance

! in ESEM and Bifactor-ESEM, all factors need to present the same patterns of freely estimated relations

! to covariates. Here, the ESEM-within-CFA approach is required to constrain the relations between the

! S-factors (but not the G-factor) and the covariates to be zero.

! In ESEM-within-CFA, one referent indicator is selected for each factor (including the G-factor) and all

! cross loadings for this indicators are constrained (@) to take exactly the value it had in the freely

! estimated model. All other loadings and cross loadings are given a start value corresponding to

! their values () from the freely estimated model. All factor variances are constrained to be 1 (@1).*

! For a bifactor model, all factor correlations are also constrained to be 0 (@0).

Model:

g BY am1@-0.39263;

g BY am5*-0.33615;

g BY am6*-0.33080;

g BY esap2@0.38216;

g BY esap4*0.33666;

g BY esav1*0.03294;

g BY emap1@0.49898;

g BY emap4*0.31362;

g BY emav4*-0.14727;

g BY inap1*0.63006;

g BY inap2*0.90622;

g BY inav1@0.57699;

g BY inav2*0.48894;

g BY ident2*0.77973;

g BY ident3*1.31872;

g BY ident4@1.28983;

g BY intrin2*1.14160;

g BY intrin4@1.14395;

g BY intrin6*1.12897;

Amotiv BY am1*0.68981;

amotiv BY am5*0.65837;

amotiv BY am6*0.65592;

amotiv BY emap1@0.22037;

amotiv BY emap4*0.19998;

amotiv BY emav4*0.00300;

amotiv BY esap2@0.15847;
amotiv BY esap4*0.17431;
amotiv BY esav1*0.11986;
amotiv BY inap1*0.05677;
amotiv BY inap2*-0.07635;
amotiv BY inav1@0.17945;
amotiv BY inav2*0.01314;
amotiv BY ident2*-0.14293;
amotiv BY ident3*0.07663;
amotiv BY ident4@0.08610;
amotiv BY intrin2@-0.06071;
amotiv BY intrin4@-0.08090;
amotiv BY intrin6*-0.07964;

extmat BY emap1*1.53774;
extmat BY emap4*0.84067;
extmat BY emav4*1.13499;
extmat BY am1@0.09013;
extmat BY am5*0.10347;
extmat BY am6*0.06301;
extmat BY esap2@0.40790;
extmat BY esap4*0.32382;
extmat BY esav1*0.45489;
extmat BY inap1*0.18847;
extmat BY inap2*0.05461;
extmat BY inav1@0.24750;
extmat BY inav2*0.06022;
extmat BY ident2*0.02634;
extmat BY ident3*0.00048;
extmat BY ident4@-0.01833;
extmat BY intrin2@-0.07773;
extmat BY intrin4@-0.04074;
extmat BY intrin6*-0.10479;

extsoc BY esap2*1.13210;
extsoc BY esap4*1.09945;
extsoc BY esav1*1.08788;
extsoc BY am1@0.08105;
extsoc BY am5*0.04667;
extsoc BY am6*0.12917;
extsoc BY emap1@0.11155;
extsoc BY emap4*0.58741;
extsoc BY emav4*0.61396;
extsoc BY inap1*0.54434;
extsoc BY inap2*0.08778;
extsoc BY inav1@0.35970;
extsoc BY inav2*0.10321;
extsoc BY ident2*-0.02858;
extsoc BY ident3*-0.06371;
extsoc BY ident4@-0.06996;

extsoc BY intrin2@-0.11342;
extsoc BY intrin4@-0.03684;
extsoc BY intrin6*-0.04847;

introj BY inap1*0.73127;
introj BY inap2*0.48896;
introj BY inav1*1.12825;
introj BY inav2*1.06249;
introj BY am1@0.01224;
introj BY am5*0.00518;
introj BY am6*0.06226;
introj BY emap1@-0.13351;
introj BY emap4*0.23147;
introj BY emav4*0.41822;
introj BY esap2@0.14887;
introj BY esap4*0.25184;
introj BY esav1*0.38897;
introj BY ident2*0.43978;
introj BY ident3*-0.16744;
introj BY ident4@0.02386;
introj BY intrin2@-0.07868;
introj BY intrin4@-0.02557;
introj BY intrin6*-0.13357;

ident BY ident2*0.37232;
ident BY ident3*0.42740;
ident BY ident4*0.59507;
ident BY am1@-0.04921;
ident BY am5*0.03338;
ident BY am6*0.01853;
ident BY emap1@-0.61895;
ident BY emap4*0.15928;
ident BY emav4*0.68637;
ident BY esap2@-0.16523;
ident BY esap4*-0.02558;
ident BY esav1*0.10667;
ident BY inap1*0.06167;
ident BY inap2*0.12079;
ident BY inav1@0.12381;
ident BY inav2*-0.03615;
ident BY intrin2@-0.00746;
ident BY intrin4@0.03462;
ident BY intrin6*0.05258;

intrin BY intrin2@0.64995;
intrin BY intrin4*0.86685;
intrin BY intrin6*0.72335;
intrin BY am1@-0.05503;
intrin BY am5*-0.00221;
intrin BY am6*-0.06204;

intrin BY emap1@-0.24644;
intrin BY emap4*-0.04727;
intrin BY emav4*0.14359;
intrin BY esap2@-0.04745;
intrin BY esap4*-0.08515;
intrin BY esav1*-0.04757;
intrin BY inap1*-0.05376;
intrin BY inap2*-0.02423;
intrin BY inav1@-0.11019;
intrin BY inav2*-0.09012;
intrin BY ident2*0.06175;
intrin BY ident3*0.06970;
intrin BY ident4@0.02096;

G-Intrin@1;
extmat WITH amotiv@0.00000;
extsoc WITH amotiv@0.00000;
extsoc WITH extmat@0.00000;
introj WITH amotiv@0.00000;
introj WITH extmat@0.00000;
introj WITH extsoc@0.00000;
ident WITH amotiv@0.00000;
ident WITH extmat@0.00000;
ident WITH extsoc@0.00000;
ident WITH introj@0.00000;
intrin WITH amotiv@0.00000;
intrin WITH extmat@0.00000;
intrin WITH extsoc@0.00000;
intrin WITH introj@0.00000;
intrin WITH ident@0.00000;
g WITH amotiv@0.00000;
g WITH extmat@0.00000;
g WITH extsoc@0.00000;
g WITH introj@0.00000;
g WITH ident@0.00000;
g WITH intrin@0.00000;

! The following inputs specify G to be freely estimated in relation to covariates while constraining

! S-factors (amotv-intrin) to be unrelated to covariates.

AC CC Autonomy Competence Relatedness ON Amotiv@0 ExtMat@0

ExtSoc@0 Introj@0 Ident@0 Intrin@0;

AC CC Autonomy Competence Relatedness ON G;

AC CC Autonomy Competence Relatedness;